

## D03PEF – NAG Fortran Library Routine Document

**Note.** Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

### 1 Purpose

D03PEF integrates a system of linear or nonlinear, first-order, time-dependent partial differential equations (PDEs) in one space variable. The spatial discretisation is performed using the Keller box scheme and the method of lines is employed to reduce the PDEs to a system of ordinary differential equations (ODEs). The resulting system is solved using a Backward Differentiation Formula (BDF) method.

### 2 Specification

```

SUBROUTINE D03PEF(NPDE, TS, TOUT, PDEDEF, BNDARY, U, NPTS, X,
1          NLEFT, ACC, W, NW, IW, NIW, ITASK, ITRACE, IND,
2          IFAIL)
  INTEGER  NPDE, NPTS, NLEFT, NW, IW(NIW), NIW, ITASK,
1          ITRACE, IND, IFAIL
  real    TS, TOUT, U(NPDE,NPTS), X(NPTS), ACC, W(NW)
  EXTERNAL PDEDEF, BNDARY

```

### 3 Description

D03PEF integrates the system of first-order PDEs

$$G_i(x, t, U, U_x, U_t) = 0, \quad i = 1, 2, \dots, \text{NPDE}. \quad (1)$$

In particular the functions  $G_i$  must have the general form:

$$G_i = \sum_{j=1}^{\text{NPDE}} P_{i,j} \frac{\partial U_j}{\partial t} + Q_i, \quad i = 1, 2, \dots, \text{NPDE}, \quad a \leq x \leq b, \quad t \geq t_0, \quad (2)$$

where  $P_{i,j}$  and  $Q_i$  depend on  $x, t, U, U_x$  and the vector  $U$  is the set of solution values

$$U(x, t) = [U_1(x, t), \dots, U_{\text{NPDE}}(x, t)]^T, \quad (3)$$

and the vector  $U_x$  is its partial derivative with respect to  $x$ . Note that  $P_{i,j}$  and  $Q_i$  must not depend on  $\frac{\partial U}{\partial t}$ .

The integration in time is from  $t_0$  to  $t_{\text{out}}$ , over the space interval  $a \leq x \leq b$ , where  $a = x_1$  and  $b = x_{\text{NPTS}}$  are the leftmost and rightmost points of a user-defined mesh  $x_1, x_2, \dots, x_{\text{NPTS}}$ . The mesh should be chosen in accordance with the expected behaviour of the solution.

The PDE system which is defined by the functions  $G_i$  must be specified in a subroutine PDEDEF supplied by the user.

The initial values of the functions  $U(x, t)$  must be given at  $t = t_0$ . For a first-order system of PDEs, only one boundary condition is required for each PDE component  $U_i$ . The NPDE boundary conditions are separated into NLEFT at the left-hand boundary  $x = a$ , and NRIGHT at the right-hand boundary  $x = b$ , such that NLEFT + NRIGHT = NPDE. The position of the boundary condition for each component should be chosen with care; the general rule is that if the characteristic direction of  $U_i$  at the left-hand boundary (say) points into the interior of the solution domain, then the boundary condition for  $U_i$  should be specified at the left-hand boundary. Incorrect positioning of boundary conditions generally results in initialisation or integration difficulties in the underlying time integration routines.

The boundary conditions have the form:

$$G_i^L(x, t, U, U_t) = 0, \quad \text{at } x = a, \quad i = 1, 2, \dots, \text{NLEFT} \quad (4)$$

at the left-hand boundary, and

$$G_i^R(x, t, U, U_i) = 0, \text{ at } x = b, \quad i = 1, 2, \dots, \text{NRIGHT} \quad (5)$$

at the right-hand boundary.

Note that the functions  $G_i^L$  and  $G_i^R$  must not depend on  $U_x$ , since spatial derivatives are not determined explicitly in the Keller box scheme [3]. If the problem involves derivative (Neumann) boundary conditions then it is generally possible to restate such boundary conditions in terms of permissible variables. Also note that  $G_i^L$  and  $G_i^R$  must be linear with respect to time derivatives, so that the boundary conditions have the general form:

$$\sum_{j=1}^{\text{NPDE}} E_{i,j}^L \frac{\partial U_j}{\partial t} + S_i^L = 0, \quad i = 1, 2, \dots, \text{NLEFT} \quad (6)$$

at the left-hand boundary, and

$$\sum_{j=1}^{\text{NPDE}} E_{i,j}^R \frac{\partial U_j}{\partial t} + S_i^R = 0, \quad i = 1, 2, \dots, \text{NRIGHT} \quad (7)$$

at the right-hand boundary, where  $E_{i,j}^L$ ,  $E_{i,j}^R$ ,  $S_i^L$ , and  $S_i^R$  depend on  $x$ ,  $t$  and  $U$  only.

The boundary conditions must be specified in a subroutine BNDARY provided by the user.

The problem is subject to the following restrictions:

- (i)  $t_0 < t_{out}$ , so that integration is in the forward direction;
- (ii)  $P_{i,j}$  and  $Q_i$  must not depend on any time derivatives;
- (iii) The evaluation of the function  $G_i$  is done at the mid-points of the mesh intervals by calling the routine PDEDEF for each mid-point in turn. Any discontinuities in the function **must** therefore be at one or more of the mesh points  $x_1, x_2, \dots, x_{\text{NPTS}}$ ;
- (iv) At least one of the functions  $P_{i,j}$  must be non-zero so that there is a time derivative present in the problem.

In this method of lines approach the Keller box scheme [3] is applied to each PDE in the space variable only, resulting in a system of ODEs in time for the values of  $U_i$  at each mesh point. In total there are  $\text{NPDE} \times \text{NPTS}$  ODEs in the time direction. This system is then integrated forwards in time using a BDF method.

## 4 References

- [1] Berzins M (1990) Developments in the NAG Library software for parabolic equations *Scientific Software Systems* (ed J C Mason and M G Cox) Chapman and Hall 59–72
- [2] Berzins M, Dew P M and Furzeland R M (1989) Developing software for time-dependent problems using the method of lines and differential-algebraic integrators *Appl. Numer. Math.* **5** 375–397
- [3] Keller H B (1970) A new difference scheme for parabolic problems *Numerical Solutions of Partial Differential Equations* (ed J Bramble) **2** Academic Press 327–350
- [4] Pennington S V and Berzins M (1994) New NAG Library software for first-order partial differential equations *ACM Trans. Math. Softw.* **20** 63–99

## 5 Parameters

- 1: NPDE — INTEGER *Input*  
*On entry:* the number of PDEs in the system to be solved.  
*Constraint:* NPDE  $\geq$  1.
- 2: TS — *real* *Input/Output*  
*On entry:* the initial value of the independent variable  $t$ .  
*Constraint:* TS < TOUT.  
*On exit:* the value of  $t$  corresponding to the solution values in U. Normally TS = TOUT.

- 3:** TOUT — *real* *Input*  
*On entry:* the final value of  $t$  to which the integration is to be carried out.
- 4:** PDEDEF — SUBROUTINE, supplied by the user. *External Procedure*  
 PDEDEF must compute the functions  $G_i$  which define the system of PDEs. PDEDEF is called approximately midway between each pair of mesh points in turn by D03PEF.  
 Its specification is:

```

SUBROUTINE PDEDEF(NPDE, T, X, U, UT, UX, RES, IRES)
INTEGER          NPDE, IRES
real           T, X, U(NPDE), UT(NPDE), UX(NPDE), RES(NPDE)

```

- 1:** NPDE — INTEGER *Input*  
*On entry:* the number of PDEs in the system.
- 2:** T — *real* *Input*  
*On entry:* the current value of the independent variable  $t$ .
- 3:** X — *real* *Input*  
*On entry:* the current value of the space variable  $x$ .
- 4:** U(NPDE) — *real* array *Input*  
*On entry:* U( $i$ ) contains the value of the component  $U_i(x, t)$ , for  $i = 1, 2, \dots, \text{NPDE}$ .
- 5:** UT(NPDE) — *real* array *Input*  
*On entry:* UT( $i$ ) contains the value of the component  $\frac{\partial U_i(x, t)}{\partial t}$ , for  $i = 1, 2, \dots, \text{NPDE}$ .
- 6:** UX(NPDE) — *real* array *Input*  
*On entry:* UX( $i$ ) contains the value of the component  $\frac{\partial U_i(x, t)}{\partial x}$ , for  $i = 1, 2, \dots, \text{NPDE}$ .
- 7:** RES(NPDE) — *real* array *Output*  
*On exit:* RES( $i$ ) must contain the  $i$ th component of  $G$ , for  $i = 1, 2, \dots, \text{NPDE}$ , where  $G$  is defined as

$$G_i = \sum_{j=1}^{\text{NPDE}} P_{i,j} \frac{\partial U_j}{\partial t}, \quad (8)$$

i.e., only terms depending explicitly on time derivatives, or

$$G_i = \sum_{j=1}^{\text{NPDE}} P_{i,j} \frac{\partial U_j}{\partial t} + Q_i, \quad (9)$$

i.e., all terms in equation (2).

The definition of  $G$  is determined by the input value of IRES.

- 8:** IRES — INTEGER *Input/Output*  
*On entry:* the form of  $G_i$  that must be returned in the array RES. If IRES = -1, then equation (8) above must be used. If IRES = 1, then equation (9) above must be used.  
*On exit:* should usually remain unchanged. However, the user may set IRES to force the integration routine to take certain actions, as described below:

IRES = 2

indicates to the integrator that control should be passed back immediately to the calling (sub)program with the error indicator (IFAIL) set to 6.

IRES = 3

indicates to the integrator that the current time step should be abandoned and a smaller time step used instead. The user may wish to set IRES = 3 when a physically meaningless input or output value has been generated. If the user consecutively sets IRES = 3, then D03PEF returns to the calling (sub)program with the error indicator (IFAIL) set to 4.

PDEDEF must be declared as EXTERNAL in the (sub)program from which D03PEF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

5: BNDARY — SUBROUTINE, supplied by the user. *External Procedure*

BNDARY must compute the functions  $G_i^L$  and  $G_i^R$  which define the boundary conditions as in equations (4) and (5).

Its specification is:

```

SUBROUTINE BNDARY(NPDE, T, IBND, NOBC, U, UT, RES, IRES)
INTEGER          NPDE, IBND, NOBC, IRES
real           T, U(NPDE), UT(NPDE), RES(NOBC)

```

1: NPDE — INTEGER *Input*  
*On entry:* the number of PDEs in the system.

2: T — **real** *Input*  
*On entry:* the current value of the independent variable  $t$ .

3: IBND — INTEGER *Input*  
*On entry:* IBND determines the position of the boundary conditions. If IBND = 0, then BNDARY must compute the left-hand boundary condition at  $x = a$ . Any other value of IBND indicates that BNDARY must compute the right-hand boundary condition at  $x = b$ .

4: NOBC — INTEGER *Input*  
*On entry:* NOBC specifies the number of boundary conditions at the boundary specified by IBND.

5: U(NPDE) — **real** array *Input*  
*On entry:* U( $i$ ) contains the value of the component  $U_i(x, t)$  at the boundary specified by IBND, for  $i = 1, 2, \dots, \text{NPDE}$ .

6: UT(NPDE) — **real** array *Input*  
*On entry:* UT( $i$ ) contains the value of the component  $\frac{\partial U_i(x, t)}{\partial t}$  at the boundary specified by IBND, for  $i = 1, 2, \dots, \text{NPDE}$ .

7: RES(NOBC) — **real** array *Output*  
*On exit:* RES( $i$ ) must contain the  $i$ th component of  $G^L$  or  $G^R$ , depending on the value of IBND, for  $i = 1, 2, \dots, \text{NOBC}$ , where  $G^L$  is defined as

$$G_i^L = \sum_{j=1}^{\text{NPDE}} E_{i,j}^L \frac{\partial U_j}{\partial t}, \quad (10)$$

i.e., only terms depending explicitly on time derivatives, or

$$G_i^L = \sum_{j=1}^{\text{NPDE}} E_{i,j}^L \frac{\partial U_j}{\partial t} + S_i^L, \quad (11)$$

i.e., all terms in equation (6), and similarly for  $G_i^R$ .

The definitions of  $G^L$  and  $G^R$  are determined by the input value of IRES.

8: IRES — INTEGER *Input/Output*  
*On entry:* the form  $G_i^L$  (or  $G_i^R$ ) that must be returned in the array RES. If IRES = -1, then equation (10) above must be used. If IRES = 1, then equation (11) above must be used.

*On exit:* should usually remain unchanged. However, the user may set IRES to force the integration routine to take certain actions, as described below:

IRES = 2

indicates to the integrator that control should be passed back immediately to the calling (sub)program with the error indicator (IFAIL) set to 6.

IRES = 3

indicates to the integrator that the current time step should be abandoned and a smaller time step used instead. The user may wish to set IRES = 3 when a physically meaningless input or output value has been generated. If the user consecutively sets IRES = 3, then D03PEF returns to the calling (sub)program with the error indicator (IFAIL) set to 4.

BNDARY must be declared as EXTERNAL in the (sub)program from which D03PEF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

**6:** U(NPDE,NPTS) — *real* array *Input/Output*  
*On entry:* the initial values of  $U(x, t)$  at  $t = TS$  and the mesh points  $X(j)$ , for  $j = 1, 2, \dots, NPTS$ .  
*On exit:*  $U(i, j)$  will contain the computed solution at  $t = TS$ .

**7:** NPTS — INTEGER *Input*  
*On entry:* the number of mesh points in the interval  $[a, b]$ .  
*Constraint:*  $NPTS \geq 3$ .

**8:** X(NPTS) — *real* array *Input*  
*On entry:* the mesh points in the spatial direction.  $X(1)$  must specify the left-hand boundary,  $a$ , and  $X(NPTS)$  must specify the right-hand boundary,  $b$ .  
*Constraint:*  $X(1) < X(2) < \dots < X(NPTS)$ .

**9:** NLEFT — INTEGER *Input*  
*On entry:* the number of boundary conditions at the left-hand mesh point  $X(1)$ .  
*Constraint:*  $0 \leq NLEFT \leq NPDE$ .

**10:** ACC — *real* *Input*  
*On entry:* a positive quantity for controlling the local error estimate in the time integration. If  $E(i, j)$  is the estimated error for  $U_i$  at the  $j$ th mesh point, the error test is:

$$|E(i, j)| = ACC \times (1.0 + |U(i, j)|).$$

*Constraint:*  $ACC > 0.0$ .

**11:** W(NW) — *real* array *Workspace*

**12:** NW — INTEGER *Input*  
*On entry:* the dimension of the array W as declared in the (sub)program from which D03PEF is called.  
*Constraint:*  $NW \geq (4 \times NPDE + NLEFT + 14) \times NPDE \times NPTS + (3 \times NPDE + 21) \times NPDE + 7 \times NPTS + 54$ .

**13:** IW(NIW) — INTEGER array *Output*  
*On exit:* the following components of the array IW concern the efficiency of the integration.

IW(1) contains the number of steps taken in time.

IW(2) contains the number of residual evaluations of the resulting ODE system used. One such evaluation involves evaluating the PDE functions at all the mesh points, as well as one evaluation of the functions in each of the boundary conditions.

IW(3) contains the number of Jacobian evaluations performed by the time integrator.

IW(4) contains the order of the last BDF method used.

IW(5) contains the number of Newton iterations performed by the time integrator. Each iteration involves an ODE residual evaluation followed by a back-substitution using the *LU* decomposition of the Jacobian matrix.

The rest of the array is used as workspace.

**14:** NIW — INTEGER *Input*

*On entry:* the dimension of the array IW as declared in the (sub)program from which D03PEF is called.

*Constraint:*  $NIW \geq NPDE \times NPTS + 24$ .

**15:** ITASK — INTEGER *Input*

*On entry:* specifies the task to be performed by the ODE integrator. The permitted values of ITASK and their meanings are described below:

ITASK = 1

normal computation of output values U at  $t = TOUT$ .

ITASK = 2

take one step and return.

ITASK = 3

stop at the first internal integration point at or beyond  $t = TOUT$ .

*Constraint:*  $1 \leq ITASK \leq 3$ .

**16:** ITRACE — INTEGER *Input*

*On entry:* the level of trace information required from D03PEF and the underlying ODE solver as follows:

If ITRACE  $\leq -1$ , no output is generated.

If ITRACE = 0, only warning messages from the PDE solver are printed on the current error message unit (see X04AAF).

If ITRACE = 1, then output from the underlying ODE solver is printed on the current advisory message unit (see X04ABF). This output contains details of Jacobian entries, the nonlinear iteration and the time integration during the computation of the ODE system.

If ITRACE = 2, then the output from the underlying ODE solver is similar to that produced when ITRACE = 1, except that the advisory messages are given in greater detail.

If ITRACE  $\geq 3$ , then the output from the underlying ODE solver is similar to that produced when ITRACE = 2, except that the advisory messages are given in greater detail.

Users are advised to set ITRACE = 0, unless they are experienced with the subchapter D02M–N of the NAG Fortran Library.

**17:** IND — INTEGER *Input/Output*

*On entry:* IND must be set to 0 or 1.

IND = 0

starts or restarts the integration in time.

IND = 1

continues the integration after an earlier exit from the routine. In this case, only the parameters TOUT and IFAIL should be reset between calls to D03PEF.

*Constraint:*  $0 \leq IND \leq 1$ .

*On exit:* IND = 1.

**18: IFAIL — INTEGER***Input/Output*

*On entry:* IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

*On exit:* IFAIL = 0 unless the routine detects an error (see Section 6).

**6 Error Indicators and Warnings**

Errors detected by the routine:

IFAIL = 1

- On entry, TOUT  $\leq$  TS,  
 or (TOUT - TS) is too small,  
 or ITASK  $\neq$  1, 2 or 3,  
 or  $X(i)$ , for  $i = 1, 2, \dots, NPTS$  are not ordered correctly,  
 or  $NPTS < 3$ ,  
 or  $NPDE < 1$ ,  
 or NLEFT is not in the range 0 to NPDE,  
 or  $ACC \leq 0.0$ ,  
 or  $IND \neq 0$  or 1,  
 or NW is too small,  
 or NIW is too small,  
 or D03PEF called initially with  $IND = 1$ .

IFAIL = 2

The underlying ODE solver cannot make any further progress, across the integration range from the current point  $t = TS$  with the supplied value of ACC. The components of U contain the computed values at the current point  $t = TS$ .

IFAIL = 3

In the underlying ODE solver, there were repeated errors or corrector convergence test failures on an attempted step, before completing the requested task. The problem may have a singularity or ACC is too small for the integration to continue. Incorrect positioning of boundary conditions may also result in this error. Integration was successful as far as  $t = TS$ .

IFAIL = 4

In setting up the ODE system, the internal initialisation routine was unable to initialise the derivative of the ODE system. This could be due to the fact that IRES was repeatedly set to 3 in the user-supplied subroutine PDEDEF or BNDARY, when the residual in the underlying ODE solver was being evaluated. Incorrect positioning of boundary conditions may also result in this error.

IFAIL = 5

In solving the ODE system, a singular Jacobian has been encountered. The user should check their problem formulation.

IFAIL = 6

When evaluating the residual in solving the ODE system, IRES was set to 2 in one of the user-supplied subroutines PDEDEF or BNDARY. Integration was successful as far as  $t = TS$ .

IFAIL = 7

The value of ACC is so small that the routine is unable to start the integration in time.

IFAIL = 8

In one of the user-supplied routines, PDEDEF or BNDARY, IRES was set to an invalid value.

IFAIL = 9

A serious error has occurred in an internal call to D02NNF. Check problem specification and all parameters and array dimensions. Setting ITRACE = 1 may provide more information. If the problem persists, contact NAG.

IFAIL = 10

The required task has been completed, but it is estimated that a small change in ACC is unlikely to produce any change in the computed solution. (Only applies when the user is not operating in one step mode, that is when ITASK  $\neq$  2.)

IFAIL = 11

An error occurred during Jacobian formulation of the ODE system (a more detailed error description may be directed to the current advisory message unit).

## 7 Accuracy

The routine controls the accuracy of the integration in the time direction but not the accuracy of the approximation in space. The spatial accuracy depends on both the number of mesh points and on their distribution in space. In the time integration only the local error over a single step is controlled and so the accuracy over a number of steps cannot be guaranteed. The user should therefore test the effect of varying the accuracy parameter, ACC.

## 8 Further Comments

The Keller box scheme can be used to solve higher-order problems which have been reduced to first-order by the introduction of new variables (see the example problem in D03PKF). In general, a second-order problem can be solved with slightly greater accuracy using the Keller box scheme instead of a finite-difference scheme (D03PCF/D03PHF for example), but at the expense of increased CPU time due to the larger number of function evaluations required.

It should be noted that the Keller box scheme, in common with other central-difference schemes, may be unsuitable for some hyperbolic first-order problems such as the apparently simple linear advection equation  $U_t + aU_x = 0$ , where  $a$  is a constant, resulting in spurious oscillations due to the lack of dissipation. This type of problem requires a discretisation scheme with upwind weighting (D03PFF for example), or the addition of a second-order artificial dissipation term.

The time taken by the routine depends on the complexity of the system and on the accuracy requested.

## 9 Example

This example is the simple first-order system

$$\begin{aligned}\frac{\partial U_1}{\partial t} + \frac{\partial U_1}{\partial x} + \frac{\partial U_2}{\partial x} &= 0, \\ \frac{\partial U_2}{\partial t} + 4\frac{\partial U_1}{\partial x} + \frac{\partial U_2}{\partial x} &= 0,\end{aligned}$$

for  $t \in [0, 1]$  and  $x \in [0, 1]$ .

The initial conditions are

$$U_1(x, 0) = \exp(x), \quad U_2(x, 0) = \sin(x),$$

and the Dirichlet boundary conditions for  $U_1$  at  $x = 0$  and  $U_2$  at  $x = 1$  are given by the exact solution:

$$\begin{aligned}U_1(x, t) &= \frac{1}{2} \{ \exp(x+t) + \exp(x-3t) \} + \frac{1}{4} \{ \sin(x-3t) - \sin(x+t) \}, \\ U_2(x, t) &= \exp(x-3t) - \exp(x+t) + \frac{1}{2} \{ \sin(x+t) + \sin(x-3t) \}.\end{aligned}$$



## 9.1 Program Text

**Note.** The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*   D03PEF Example Program Text
*   Mark 16 Release. NAG Copyright 1993.
*   .. Parameters ..
INTEGER          NOUT
PARAMETER       (NOUT=6)
INTEGER         NPDE, NPTS, NLEFT, NEQN, NIW, NWKRES, NW
PARAMETER       (NPDE=2,NPTS=41,NLEFT=1,NEQN=NPDE*NPTS,
+              NIW=NEQN+24,NWKRES=NPDE*(NPTS+21+3*NPDE)
+              +7*NPTS+4,NW=11*NEQN+(4*NPDE+NLEFT+2)
+              *NEQN+50+NWKRES)
*   .. Local Scalars ..
  real         ACC, TOUT, TS
INTEGER         I, IFAIL, IND, IT, ITASK, ITRACE
*   .. Local Arrays ..
  real         EU(NPDE,NPTS), U(NPDE,NPTS), W(NW), X(NPTS)
INTEGER         IW(NIW)
*   .. External Subroutines ..
EXTERNAL        BNDARY, D03PEF, EXACT, PDEDEF, UINIT
*   .. Executable Statements ..
WRITE (NOUT,*) 'D03PEF Example Program Results'
ITRACE = 0
ACC = 0.1e-5
WRITE (NOUT,99997) ACC, NPTS
*
*   Set spatial-mesh points
*
DO 20 I = 1, NPTS
  X(I) = (I-1.0e0)/(NPTS-1.0e0)
20 CONTINUE
WRITE (NOUT,99999) X(5), X(13), X(21), X(29), X(37)
*
IND = 0
ITASK = 1
*
CALL UINIT(NPDE,NPTS,X,U)
*
*   Loop over output value of t
TS = 0.0e0
TOUT = 0.0e0
DO 40 IT = 1, 5
  TOUT = 0.2e0*IT
  IFAIL = -1
*
  CALL D03PEF(NPDE,TS,TOUT,PDEDEF,BNDARY,U,NPTS,X,NLEFT,ACC,W,NW,
+           IW,NIW,ITASK,ITRACE,IND,IFAIL)
*
*   Check against the exact solution
*
CALL EXACT(TOUT,NPDE,NPTS,X,EU)
*
WRITE (NOUT,99998) TS
WRITE (NOUT,99995) U(1,5), U(1,13), U(1,21), U(1,29), U(1,37)
WRITE (NOUT,99994) EU(1,5), EU(1,13), EU(1,21), EU(1,29),
+   EU(1,37)

```

```

        WRITE (NOUT,99993) U(2,5), U(2,13), U(2,21), U(2,29), U(2,37)
        WRITE (NOUT,99992) EU(2,5), EU(2,13), EU(2,21), EU(2,29),
+         EU(2,37)
40 CONTINUE
        WRITE (NOUT,99996) IW(1), IW(2), IW(3), IW(5)
        STOP
*
99999 FORMAT (' X          ',5F10.4,/)
99998 FORMAT (' T = ',F5.2)
99997 FORMAT (//' Accuracy requirement =',e10.3,' Number of points = ',
+         I3,/)
99996 FORMAT (' Number of integration steps in time = ',I6,/' Number o',
+         'f function evaluations = ',I6,/' Number of Jacobian eval',
+         'uations = ',I6,/' Number of iterations = ',I6,/)
99995 FORMAT (' Approx U1',5F10.4)
99994 FORMAT (' Exact U1',5F10.4)
99993 FORMAT (' Approx U2',5F10.4)
99992 FORMAT (' Exact U2',5F10.4,/)
        END
*
        SUBROUTINE UINIT(NPDE,NPTS,X,U)
*       Routine for PDE initial values
*       .. Scalar Arguments ..
        INTEGER          NPDE, NPTS
*       .. Array Arguments ..
        real             U(NPDE,NPTS), X(NPTS)
*       .. Local Scalars ..
        INTEGER          I
*       .. Intrinsic Functions ..
        INTRINSIC        EXP, SIN
*       .. Executable Statements ..
        DO 20 I = 1, NPTS
            U(1,I) = EXP(X(I))
            U(2,I) = SIN(X(I))
20 CONTINUE
        RETURN
        END
*
        SUBROUTINE PDEDEF(NPDE,T,X,U,UDOT,DUDX,RES,IRES)
*       .. Scalar Arguments ..
        real             T, X
        INTEGER          IRES, NPDE
*       .. Array Arguments ..
        real             DUDX(NPDE), RES(NPDE), U(NPDE), UDOT(NPDE)
*       .. Executable Statements ..
        IF (IRES.EQ.-1) THEN
            RES(1) = UDOT(1)
            RES(2) = UDOT(2)
        ELSE
            RES(1) = UDOT(1) + DUDX(1) + DUDX(2)
            RES(2) = UDOT(2) + 4.0e0*DUDX(1) + DUDX(2)
        END IF
        RETURN
        END
*

```

```

SUBROUTINE BNDARY(NPDE,T,IBND,NOBC,U,UDOT,RES,IRES)
*
.. Scalar Arguments ..
real          T
INTEGER        IBND, IRES, NOBC, NPDE
*
.. Array Arguments ..
real          RES(NOBC), U(NPDE), UDOT(NPDE)
*
.. Intrinsic Functions ..
INTRINSIC      EXP, SIN
*
.. Executable Statements ..
IF (IBND.EQ.0) THEN
  IF (IRES.EQ.-1) THEN
    RES(1) = 0.0e0
  ELSE
    RES(1) = U(1) - 0.5e0*(EXP(T)+EXP(-3.0e0*T)) -
+           0.25e0*(SIN(-3.0e0*T)-SIN(T))
  END IF
ELSE
  IF (IRES.EQ.-1) THEN
    RES(1) = 0.0e0
  ELSE
    RES(1) = U(2) - EXP(1.0e0-3.0e0*T) + EXP(1.0e0+T) -
+           0.5e0*(SIN(1.0e0-3.0e0*T)+SIN(1.0e0+T))
  END IF
END IF
RETURN
END

*
SUBROUTINE EXACT(T,NPDE,NPTS,X,U)
*
Exact solution (for comparison purposes)
*
.. Scalar Arguments ..
real          T
INTEGER        NPDE, NPTS
*
.. Array Arguments ..
real          U(NPDE,NPTS), X(NPTS)
*
.. Local Scalars ..
INTEGER        I
*
.. Intrinsic Functions ..
INTRINSIC      EXP, SIN
*
.. Executable Statements ..
DO 20 I = 1, NPTS
  U(1,I) = 0.5e0*(EXP(X(I)+T)+EXP(X(I)-3.0e0*T)) +
+         0.25e0*(SIN(X(I)-3.0e0*T)-SIN(X(I)+T))
  U(2,I) = EXP(X(I)-3.0e0*T) - EXP(X(I)+T) + 0.5e0*(SIN(X(I)
+         -3.0e0*T)+SIN(X(I)+T))
20 CONTINUE
RETURN
END

```

## 9.2 Program Data

None.

### 9.3 Program Results

D03PEF Example Program Results

Accuracy requirement = 0.100E-05 Number of points = 41

X            0.1000    0.3000    0.5000    0.7000    0.9000

T = 0.20

Approx U1    0.7845    1.0010    1.2733    1.6115    2.0281

Exact U1    0.7845    1.0010    1.2733    1.6115    2.0281

Approx U2   -0.8352   -0.8159   -0.8367   -0.9128   -1.0609

Exact U2   -0.8353   -0.8160   -0.8367   -0.9129   -1.0609

T = 0.40

Approx U1    0.6481    0.8533    1.1212    1.4627    1.8903

Exact U1    0.6481    0.8533    1.1212    1.4627    1.8903

Approx U2   -1.5216   -1.6767   -1.8934   -2.1917   -2.5944

Exact U2   -1.5217   -1.6767   -1.8935   -2.1917   -2.5945

T = 0.60

Approx U1    0.6892    0.8961    1.1747    1.5374    1.9989

Exact U1    0.6892    0.8962    1.1747    1.5374    1.9989

Approx U2   -2.0047   -2.3434   -2.7677   -3.3002   -3.9680

Exact U2   -2.0048   -2.3436   -2.7678   -3.3003   -3.9680

T = 0.80

Approx U1    0.8977    1.1247    1.4320    1.8349    2.3514

Exact U1    0.8977    1.1247    1.4320    1.8349    2.3512

Approx U2   -2.3403   -2.8675   -3.5110   -4.2960   -5.2536

Exact U2   -2.3405   -2.8677   -3.5111   -4.2961   -5.2537

T = 1.00

Approx U1    1.2470    1.5206    1.8828    2.3528    2.9519

Exact U1    1.2470    1.5205    1.8829    2.3528    2.9518

Approx U2   -2.6229   -3.3338   -4.1998   -5.2505   -6.5218

Exact U2   -2.6232   -3.3340   -4.2001   -5.2507   -6.5219

Number of integration steps in time = 149

Number of function evaluations = 399

Number of Jacobian evaluations = 13

Number of iterations = 323